

```
*-----;
* 昭和大学 第4回実践臨床統計学セミナー・ハンズオン2 ;
* 対応のあるデータと3群以上の比較 ;
* 2020年2月3日 ;
*-----;
```

```
*-----;
* 1. CSVデータファイルの読み込み ;
*-----;
```

```
proc import out=work.intake
datafile = "C:\SAS2\intake.csv"
dbms = csv replace;
getnames = yes;
datarow = 2;
run;
```

\*-- datafile=... のフォルダ・ファイルのパスは適当に変更してください。;  
\*-- Cドライブ直下に "SAS2" というフォルダを作ってくださいと、以後の実習でも、プログラムの改変なしに、配布したコードでデモを実行することができます。;

```
*-----;
* 2. TTESTプロシジャ： 対応のある一標本 t 検定を行います ;
*-----;
```

```
proc ttest data=intake;
paired pre * post ;
run;
```

\*-- PAIREDステートメントを入れることで、対応のあるt検定を行うことができます。;

```
*-----;
* 3. UNIVARIATEプロシジャ： 対応のある一標本Wilcoxon検定を行います ;
*-----;
```

```
data intake2;
set intake;
diff = post-pre;
run;
```

\*-- 前後値の差をとった変数 **diff** を追加したデータセット **intake2** を作っています。;  
\*-- 対応のある一標本検定は、「diffの平均値が0」という帰無仮説の検定になります。;

```
proc univariate data=intake2;
var diff;
run;
```

\*--  $\mu_0=0$ の検定が、対応のある一標本検定の結果になります。;  
\*-- NPAR1WAYプロシジャは、対応のある検定には対応していないので、注意。;

```
*-----;
* 4. CSVデータファイルの読み込み ;
*-----;
```

```
proc import out=work.ashina
datafile = "C:\SAS2\ashina.csv"
dbms = csv replace;
getnames = yes;
datarow = 2;
run;
```

```
*-----;
* 5. TTESTプロシジャ： 対応のある一標本 t 検定を行います ;
*-----;
```

```
proc ttest data=ashina;
paired vas_plac * vas_active ;
run;
```

```
*-----;
* 6. CSVデータファイルの読み込み ;
*-----;
```

```
proc import out=work.cereb
datafile = "C:\SAS2\cereb.csv"
dbms = csv replace;
getnames = yes;
datarow = 2;
run;
```

```
*-----;
* 7. FREQプロシジャ： McNamer検定を行います ;
*-----;
```

```
proc freq data=cereb;
tables y0*y1 /agree expected norow nocol nopercnt;
run;
```

```
*-----;
* 8. CSVデータファイルの読み込み ;
*-----;
```

```
proc import out=work.gold
datafile = "C:\SAS2\gold.csv"
dbms = csv replace;
getnames = yes;
datarow = 2;
run;
```

```
*-----;
* 9. FREQプロシジャ： McNamer検定を行います ;
*-----;
```

```
proc freq data=gold;
tables case*control /agree expected norow nocol nopercnt relrisk;
run;
```

\*-- FREQプロシジャのrelriskで計算されるオッズ比などは、対応（マッチング）を考慮していないので、誤ったものとなっています。；  
\*-- ここで出てくるオッズ比と信頼区間を、論文などに採用してはいけないので、注意。；

```
*-----;
* 10. CSVデータファイルの読み込み ;
*-----;
```

```
proc import out=work.gold2
datafile = "C:\SAS2\gold2.csv"
dbms = csv replace;
getnames = yes;
datarow = 2;
run;
```

```
*-----;
* 11. FREQプロシジャ： Cochran-Mantel-Haenszel推定量の計算を行います ;
*-----;
```

```
proc freq data=gold2;
table pairs*y*x / cmh noprint;
run;
```

\*-- cmhオプションをつけることで、オッズ比のCochran-Mantel-Haenszel推定量を計算することができます。；  
\*-- マッチングをとった研究でのオッズ比の推定値・信頼区間は、この方法で計算したものを使いましょう。；

```
*-----;
* 12. CSVデータファイルの読み込み ;
*-----;
```

```
proc import out=work.rcf
datafile = "C:\SAS2\red.cell.folate.csv"
dbms = csv replace;
getnames = yes;
datarow = 2;
run;
```

```
*-----;
* 13. GLMプロシジャ：一元配置分散分析を行います ;
*-----;
```

```
proc means data=rcf nonobs n mean stddev min max fw=6;
class ventilation;
var folate;
run;
```

\*-- MEANSプロシジャで、群ごとのアウトカム変数の要約統計量をまとめて計算することができます。;

```
proc glm data=rcf;
class ventilation;
model folate=ventilation;
means ventilation;
run;
```

\*-- GLMプロシジャで、一元配置分散分析を行うことができます。;

\*-- MEANSステートメントで指定した変数について、平均・標準偏差といった要約統計量を一緒に出力することができます。;

```
*-----;
* 14. NPAR1WAYプロシジャ：Kruskal-Wallis検定を行います ;
*-----;
```

```
proc npar1way data=rcf wilcoxon;
class ventilation;
var folate;
run;
```

```
*-----;
* 15. 分割表データの直接入力と分割表の検定 ;
*-----;
```

```
data actg;
input trt $ event $ n;
cards;
ZID 1 181
ZID_DID 1 103
ZID_ZAL 1 109
DID 1 128
ZID 0 351
ZID_DID 0 419
ZID_ZAL 0 415
DID 0 433
;
run;
```

\*-- 分割表のデータなどは、セル度数がわかっているものであれば、このように直接入力することができます。;

```
proc freq data=actg;
tables trt*event/ nopercnt nocol chisq;
weight n;
```

run;

- \*-- **FREQ**プロシジャで、**3群**以上のイベント発生率に差があるかどうかのカイ二乗検定を行うことができます。;
- \*-- 上記のデータ形式だと、"**weight n**"を入れる必要があるので注意。;